
TAMPILKAN NADA DTMF DAN DERING TELEPHONE 
OLEH MODUL DF-88 DAN MODUL DST-51 PADA LCD 

 
 Pada aplikasi-aplikasi menggunakan saluran telephone, proses deteksi nada 
DTMF maupun sinyal dering seringkali dibutuhkan, oleh karena itu di sini akan 
dibahas bagaimana proses mendeteksi nada DTMF dengan menggunakan IC MT8888 
yang terdapat dalam Modul DF-88 sekaligus mendeteksi nada dering. Nada DTMF 
maupun indikator dering ditampilkan pada LCD 16x2 dengan standard M1632. 
 Selain untuk mempelajari proses pengambilan Nada DTMF dan sinyal dering, 
dengan menambahkan program jam digital yang dibahas pada topik Antar Muka 
Modul Real Time Clock dengan Modul DST-51 pada edisi 297 ke dalam program 
yang dibahas pada aplikasi ini, maka aplikasi ini akan menjadi aksesoris tambahan 
yang cukup menarik pada pesawat telephone anda. LCD akan selalu menampilkan 
jam pada baris pertama dari layarnya sedangkan Nada DTMF atau pesan “RING” 
akan tampil pada baris kedua setiap kali sistem ini mendeteksi adanya nada DTMF 
ataupun sinyal dering. 
 
Cara Kerja 
 Bagian perangkat keras terdiri dari 3 bagian yaitu Detektor DTMF, Detektor 
Ring dan Antar Muka LCD. Bagian Detektor DTMF terdiri dari IC MT8888 yang 
berfungsi untuk mengubah nada DTMF menjadi biner dan trafo AX141 yang 
berfungsi sebagai coupling antara saluran telephone dan rangkaian. 

Skema Detektor Nada DTMF 

 
 MT8888 sebetulnya tidak hanya berfungsi sebagai detektor nada DTMF saja, 
namun pada aplikasi lain dapat juga digunakan sebagai pengirim nada DTMF (DTMF 
Encoder) maupun Call Progress Detector. 
 Bagian Ring Detektor terdiri dari opto coupler H11AA1 di mana pada saat 
sinyal dering masuk melalui J1 (konektor ke saluran telephone) menuju ke J2 
(konektor ke pesawat telephone), maka opto coupler H11AA1 akan mengubah sinyal 
dering yang berbentuk sinus 50 Hz dengan amplitudo 40 Volt RMS menjadi 
gelombang kotak pada titik hook di mana besarannya dapat diukur oleh 
mikrokontroler 



 
 

Deteksi Sinyal Dering 
 

 
Gambar Sinyal Dering 

 
 Selain berfungsi sebagai detektor sinyal dering, bagian ini juga berfungsi 
sebagai detektor hook (gagang) yang mendeteksi gagang diangkat (OFF Hook) dan 
gagang ditutup (ON Hook). Pada saat gagang telephone diangkat (Off Hook) maka 
impedansi pesawat telephone akan turun menjadi 600 hingga 1K ohm sehingga arus 
akan mengalir dari J1 menuju J2 melalui opto coupler H11AA1 atau sebaliknya dari 
J1, J2 ke opto coupler H11AA1. Hal ini tergantung dari posisi TIP dan RING yang 
terhubung pada J1 

Aliran arus ini akan menyebabkan opto coupler saturasi dan tegangan di titik 
hook akan turun menjadi 0 Volt yang menandakan bahwa posisi gagang telephone 
pada off hook.  
Pada saat gagang telephone berada pada posisi on hook maka impedansi pesawat 
telephone akan kembali tinggi sehingga arus tidak akan mengalir melalui opto 
coupler. Hal ini menyebabkan opto coupler cut off dan tegangan pada titik hook 
berubah menjadi 5 Volt. 
 



Pada saat sinyal dering masuk melalui J1, maka opto coupler H11AA1 akan 
mengubah sinyal tersebut menjadi gelombang kotak pada titik hook di mana 
besarannya dapat diukur oleh microcontroller 
 

 
Aliran Arus pada Detektor Hook 

 
 Bagian antar muka modul LCD terhubung ke AT89C51 dengan teknik antar 
muka 4 bit di mana setiap 8 bit data yang dikirimkan dua kali secara bergantian 
dengan diiringi sebuah pulsa pada kaki E. 

Antar Muka LCD 

 
 

Bagian perangkat lunak di sini terdiri dari program utama, rutin interupsi 
DTMF dan rutin interupsi dering. Proses yang dilakukan pada program utama 
hanyalah berupa inisialisasi-inisialisasi saja yaitu, inisialisasi perangkat lunak untuk 
deteksi dering, inisialisasi MT8888 dan inisialisasi LCD. Selanjutnya program akan 
berhenti di alamat tersebut tanpa melakukan aksi apapun.  



Inisial Software untuk

Aktifkan Interupsi

Program Utama 
 

Aksi-aksi hanya dilakukan setelah terjadi interupsi baik interupsi yang 
ditimbulkan oleh sinyal DTMF ataupun interupsi yang ditimbulkan oleh sinyal dering. 
Pada bagian inilah anda dapat menambahkan program jam digital sehingga program 
utama akan selalu menampilkan jam pada LCD saat nada DTMF dan sinyal dering 
tidak terdeteksi. 

Saat sinyal dering terdeteksi gelombang kotak dari rangkaian detektor hook 
yang terhubung ke kaki T0 dari AT89C51 dalam modul DST-51 akan menyebabkan 
counter 0 AT89C51 menghitung hingga meluap (overflow). Hal ini disebabkan karena 
pada saat proses inisialisasi software untuk deteksi ring, register TMOD diinisial 
dengan Timer 0, mode counter. 
 
Potongan Program Inisial Ring Detektor 
BanyakPulsa_Ring EQU -40 
Init_Ring: 
 Clr TF0    ;Hapus Flag Timer 0 
 Mov TMOD,#05H   ;Timer 0 Mode 16 bit counter 
 Lcall Reset_Counter0  ;Set Counter 0 -40 
 Setb TR0    ;Start Timer 0 
 Setb ET0    ;Aktifkan Timer 0 Interrupt 
 Setb EA    ;Aktifkan interupsi semua 
 Ret 
Reset_Counter0: 
 Mov TH0,#0FFH 
 Mov TL0,#BanyakPulsa_Ring 
 Ret 

 
 Nilai dari counter 0 yaitu TH0 dan TL0 diisi dengan –40 sehingga counter 
akan meluap saat berubah menjadi 00 setelah 40 kali hitungan atau 40 pulsa dering. 
Nada dering terdengar selama 2 detik dengan frekwensi 50 Hz, oleh karena itu setiap 
nada dering terdiri dari 100 kali getaran sinyal. Nilai counter 0 dapat diatur antara –12 
yaitu jumlah getaran maksimal sinyal pemutaran mode pulse (pulse mode dialing) dan 
– 100. 
 Setelah sinyal dering bergetar 40 kali (sesuai yang diset pada program ini) 
maka counter 0 akan meng-interupsi program dan menjalankan rutin interupsi counter 
0. Pada rutin ini, pesan “RING” ditampilkan pada LCD dengan terlebih dahulu 
mematikan interupsi counter 0 dan menghentikan counter tersebut. 

Rutin Interupsi Counter 0 

 
 Setelah ditunda selama 1 detik, counter 0 kembali direset dan tampilan di layar 
LCD dihapus. Program kembali melompat ke program utama dan berhenti di sana 
hingga interupsi selanjutnya dari nada dering berikut atau nada DTMF terdengar. 
 Nada DTMF terdeteksi dengan adanya kondisi logika 0 pada INT0 yang 
terhubung pada kaki IRQ MT8888. Kondisi ini menimbulkan interupsi pada program 
yang berjalan dalam AT89C51 dan mengakibatkan program melompat ke vektor 
interupsi eksternal 0.  
 Hasil konversi nada DTMF diambil di Register Data dari IC MT8888 dalam 
rupa biner dan dikirim ke LCD dalam bentuk ASCII. Namun sesuai dengan mode 
pemutaran pulse, angka 0 terdiri dari 10 putaran, maka dalam bentuk DTMF angka 0 
diwakili dengan data 10 atau 0AH. Agar dapat dikirim ke LCD dengan benar, maka 
saat data yang diperoleh adalah 10 maka data tersebut harus diubah menjadi 0 terlebih 
dahulu. 
 Kemudian agar tidak terjadi efek bouncing, program pada mikrokontroler 
memastikan apakah nada DTMF masih terdengar dengan membaca isi register status. 
Setelah nada DTMF tidak terdengar dilanjutkan dengan memastikan apakah interupsi 
DTMF masih aktif. Apabila interupsi ini masih aktif, maka program akan menunggu 
hingga interupsi ini tidak lagi aktif dan dilanjutkan dengan deteksi kondisi gagang. 
 Saat gagang masih terangkat, program masih bekerja dalam rutin ini untuk 
menunggu nada DTMF selanjutnya. Namun bila gagang sudah ditutup maka program 
akan menghapus layar LCD dan kembali ke program utama di lokasi saat terjadinya 
interupsi. Proses deteksi gagang tidak hanya dilakukan di sini, namun pada saat 
program menunggu nada DTMF hilang dan menunggu interupsi non aktif, proses 
deteksi gagang juga dilakukan. 
 Listing program berikut ini dibangun dengan menggunakan rutin-rutin yang 
ada dalam Modul DST-51 yang juga dapat diambil dari www.delta-electronic.com 
bagian software. 

Rutin Interupsi DTMF 

 
*********************************************************************
********* 
* PROGRAM PENAMPIL JAM DENGAN RTC 1287 DAN DTMF DENGAN MODUL DF-
88         
*  - Menampilkan jam setiap detik ke serial port DST-51       
*  - Menampilkan nada DTMF ke LCD setiap penekanan tombol 
keypad 
*  - Menampilkan pesan "ring" setiap sinyal dering 
* Disain oleh: 
* Paulus Andi Nalwan, ST 
* electronic@sby.centrin.net.id 



* 
* DELTA ELECTRONIC 
* Kompleks Ruko Manyar Megah Indah Plaza D-22 
* Ngagel Jaya Selatan 
* Surabaya 60284 
* Phone/fax: 62-31-5020210 
* http://www.delta-electronic.com 
*********************************************************************
********* 
 .CODE 
;------------- 
;Deklarasi I/O 
;------------- 
Gagang   Bit T0  ;Detektor Gagang telephone 
      ;dan ring 
InterupsiDTMF  Bit INT0  ;Detektor interupsi DTMF 
 
;------------- 
;Konstanta yang digunakan 
;------------- 
BanyakPulsa_Ring EQU -40  ;Jumlah sinyal dalam 1x ring 
Alamat_DF88 EQU 8000H   ;Alamat Modul DF88 
 
;------------- 
;Rutin DST-51 yang digunakan 
;------------- 
Delay_1detik  EQU 015CH ;Delay 1 detik 
PosisiAwal_LCD  EQU 0216H ;cursor LCD pindah ke posisi awal 
Init_LCD  EQU 0237H ;Inisialisasi LCD 
KirimPesan_LCD  EQU 022BH ;Kirim pesan yang ditunjuk DPTR 
ke 
     ;LCD 
Kirim_Karakter  EQU 0295H ;Kirim Akumulator ke LCD 
Baris2_LCD  EQU 029FH ;Cursor LCD pindah ke baris 2 
Init_MT8888  EQU 0682H ;Reset MT8888 
 
 
 Org 2000H   ;Vektor Reset 
 Ajmp Start   ; 
 Org 2003H   ;Vektor Interupsi Eksternal 0 
 Ajmp Int0 
 Org 200BH   ;Vektor Interupsi Timer 0 
 Ajmp Ring 
 Org 2013H   ;Vektor Interupsi Eksternal 1 
 Reti 
 Org 201BH   ;Vektor Interupsi Timer 1 
 Reti    ; 
 Org 2023H   ;Vektor Interupsi Serial 
 Reti    ; 
 
Start: 
 Acall Init_Ring  ;Inisial Software untuk Ring Detektor 
 Mov DPTR,#Alamat_DF88 ;Reset MT8888 
 Lcall Init_Mt8888  ; 
 Setb EX0   ;Aktifkan Interupsi DTMF 
 Lcall Init_LCD 
 Ajmp * 
 
 
Init_Ring: 
 Clr TF0    ;Hapus Flag Timer 0 



 Mov TMOD,#05H   ;Timer 0 Mode 16 bit counter 
 Lcall Reset_Counter0   ;Set Counter 0 -40 
 Setb TR0    ;Start Timer 0 
 Setb ET0    ;Aktifkan Timer 0 Interrupt 
 Setb EA    ;Aktifkan interupsi semua 
 Ret 
 
Reset_Counter0: 
 Mov TH0,#0FFH 
 Mov TL0,#BanyakPulsa_Ring 
 Ret 
 
***** Subroutine untuk menampilkan pesan 'ring' di LCD setiap ring 
terjadi *** 
 
Ring: 
 Clr EA    ;Matikan Seluruh interupsi 
 Clr ET0    ; 
 Clr TR0    ;Matikan Counter 0 
 Push DPH    ;Simpan DPTR di Stack 
 Push DPL    ; 
 Lcall Baris2_LCD 
 Mov DPTR,#Pesan_Ring  ;Tampilkan Pesan Ring 
 Lcall KirimPesan_LCD   ; 
 Lcall Reset_Counter0   ; 
 Lcall Delay_1detik 
 Lcall Init_LCD 
 Pop DPL    ;Ambil DPTR dari Stack 
 Pop DPH    ; 
 Setb TR0    ;Aktifkan Counter 0 
 Setb ET0    ;Aktifkan interupsi counter 0 
 Setb EA    ;Aktifkan Seluruh Interrupt 
 Setb C    ;Set Carry Flag 
 Reti 
 
Pesan_Ring: 
 DB 'Ring',0FH 
 
INT0: 
 Push DPH    ;Simpan register2 yg digunakan 
 Push DPL    ;dalam rutin ini ke stack 
 Push A    ; 
 Clr EX0    ;Matikan Interupsi 0 
 Lcall Baris2_LCD   ;Cursor pindah ke baris 2 
LoopDTMF: 
 Mov DPTR,#Alamat_DF88  ;Ambil Data DTMF dari register 
 Movx A,@DPTR    ;data MT8888 
 Anl A,#0FH    ; 
 Cjne A,#10,BukanNol   ;Konversi 10 ke nol 
 Clr A    ; 
BukanNol: 
 Add A,#30H    ;Ubah ke ASCII Numerik 
 Lcall Kirim_Karakter   ;Kirim ke LCD 
 Mov DPTR,#Alamat_DF88+100H  ; 
 
;--------- 
;Bagian ini berfungsi untuk menunggu sinyal DTMF hilang sambil 
memeriksa 
;kondisi Hook 
TungguDTMF: 
 Movx A,@DPTR     



 Jb Gagang,ONHook   ;Bila ON Hook, lompat ke  
      ;label ONHook 
 Jnb A.3,TungguDTMF    
 Mov DPTR,#Alamat_DF88+100H   
;--------- 
;Bagian ini berfungsi untuk menunggu interupsi MT8888 non aktif 
sambil 
;memeriksa kondisi Hook 
TungguIRQ: 
 Movx A,@DPTR 
 Jb Gagang,ONHook 
 Jnb A.0,TungguIRQ 
 Jnb INT0,TungguIRQ 
;---------- 
 
 Jnb Gagang,LoopDTMF  ;Periksa kondisi hook 
ONHook: 
 Lcall Init_LCD  ;ON Hook, hapus layar LCD 
 Pop A   ;Ambil register-register yang disim 
 Pop DPL   ;pan di stack 
 Pop DPH   ; 
 Setb EX0   ;Aktifkan interupsi DTMF 
 Reti 

 


